
* This do file replicates Figure 4, about historical family types and labor market indicators

/* Structure of the do file:
1. Append micro data from EPA for each election year
2. Generate relevant variables for the analysis
3. Collapse individual-level data at the province-year-gender level
*/

	
**1. # We append all datasets

	use "Datasets/00_rawdata/epa_1979_q1_replication.dta", clear

	append using "Datasets/00_rawdata/epa_1979_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_1983_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_1983_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_1987_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_1987_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_1991_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_1991_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_1995_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_1995_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_1999_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_1999_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_2003_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_2003_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_2007_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_2007_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_2011_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_2011_q2_replication.dta"
	append using "Datasets/00_rawdata/epa_2015_q1_replication.dta"
	append using "Datasets/00_rawdata/epa_2015_q2_replication.dta"

**# 2. We generate relevant variables for the analysis
	
	gen period=yq(year, quarter)
	format period %tq

	gen edad=.
	replace edad=1 if EDAD>=16 & EDAD<=24		/*16-24*/
	replace edad=2 if EDAD>=25 & EDAD<=34 		/*25-34 */
	replace edad=3 if EDAD>=35 & EDAD<=44		/*35-44 */
	replace edad=4 if EDAD>=45 & EDAD<=54		/*45-54*/
	replace edad=5 if EDAD>=55 & EDAD<=64		/*55-64*/
	replace edad=6 if EDAD>=65 					/*65 or more*/
	replace edad=. if EDAD==.    
	
	replace edad=1 if EDAD5==16 | EDAD5==20 	/*16-24*/
	replace edad=2 if EDAD5==25 | EDAD5==30		/*25-34 */
	replace edad=3 if EDAD5==35 | EDAD5==40 	/*35-44 */
	replace edad=4 if EDAD5==45 | EDAD5==50		/*45-54*/
	replace edad=5 if EDAD5==55 | EDAD5==60		/*55-64*/
	replace edad=6 if EDAD5==65					/*65 or more*/
	/*there are no missing values in EDAD5*/
	
	* this is for year 2015 (all variables are lower case)
	replace edad=1 if edad5==16 | edad5==20 	/*16-24*/
	replace edad=2 if edad5==25 | edad5==30		/*25-34 */
	replace edad=3 if edad5==35 | edad5==40 	/*35-44 */
	replace edad=4 if edad5==45 | edad5==50		/*45-54*/
	replace edad=5 if edad5==55 | edad5==60		/*55-64*/
	replace edad=6 if edad5==65					/*65 o mas*/
	/*there are no missing values in edad5*/
	
	gen ageg=.						
	replace ageg=1 if edad==1				/*16-24*/
	replace ageg=2 if edad==2 | edad==3		/*25-44*/
	replace ageg=3 if edad==4 | edad==5 	/*45-65*/
	
* We drop observations for which we miss the age
	drop if ageg==.

	gen woman=.
	replace woman=1 if SEXO==6
	replace woman=0 if SEXO==1
	replace woman=1 if SEXO1==6
	replace woman=0 if SEXO1==1
	replace woman=1 if sexo1==6
	replace woman=0 if sexo1==1
	
	rename PROV provin
	
	
* SILA is for years 1979, 1983, 1987 and 1991
capture drop work
gen work=0
replace work=1 if SILA==2 | SILA==3 | SILA==4 | SILA==5
 /*2: realizo algun trabajo en su ocupacion principal,
3: realizo algun trabajo en su ocupacion secundaria, 4: trabajaba en el extrangero, 5: tenia empleo y no trabajo*/

capture drop unemp /*unemployed*/
gen unemp=0
replace unemp=1 if SILA==6
/*6: no trabajo, esta disponible y busca empleo*/
capture drop inact /*inactive*/
gen inact=0
replace inact=1 if SILA==7 | SILA==8
/* 7: no trabajo, esta disponible y no busca empleo, 8: otra situacion*/

* AOI1 is for years 1995 only
replace work=1 if (AOI1==1 | AOI1==3 | AOI1==4) & year==1995 
/*1: poblacion contada aparte que trabaja, 3: ocupados subempleados, 4: resto de ocupados*/
replace unemp=1 if (AOI1==5 | AOI1==6) & year==1995 
/*5: parados que buscan primer empleo, 6: parados que han trabajado antes */
replace inact=1 if (AOI1==7 | AOI1==8 | AOI1==9) & year==1995 
/* 7: inactivos (desanimados), 8: inactivos 2, 9: inactivos 3*/
	
* AOI is for years 1999, 2003, 2007 and 2011
replace work=1 if (AOI==1 |  AOI==3 | AOI==4) & (year==1999 | year==2003 | year==2007 | year==2011)  
/*1: poblacion contada aparte que trabaja, 3: ocupados subempleados por insuficiencia de horas, 4: resto de ocupados*/
replace unemp=1 if (AOI==5 | AOI==6) & (year==1999 | year==2003 | year==2007 | year==2011)  
/*5: parados que buscan primer empleo, 6: parados que han trabajado antes */
replace inact=1 if (AOI==7 | AOI==8 | AOI==9) & (year==1999 | year==2003 | year==2007 | year==2011)  
/* 7: inactivos (desanimados), 8: inactivos 2 (junto con inactivos 1 forman los activos potenciales, 9: inactivos 3*/

* aoi is for year 2015
replace work=1 if (aoi==3 | aoi==4)  & year==2015
/* 3: ocupados subempleados por insuficiencia de horas, 4: resto de ocupados*/
replace unemp=1 if (aoi==5 | aoi==6)  & year==2015
/*5: parados que buscan primer empleo, 6: parados que han trabajado antes */
replace inact=1 if (aoi==7 | aoi==8 | aoi==9)  & year==2015
/* 7: inactivos (desanimados), 8: inactivos 2 (junto con inactivos 1 forman los activos potenciales, 9: inactivos 3 (resto de inactivos)*/
	
cap drop active /*active population, all except "inactivo"*/
gen active=0
replace active=1 if work==1 | unemp==1

* we drop ceuta and melilla (Spanish territories in northern Africa)
drop if provin==51
drop if provin==52

**# 3. We collapse the individual-level data at the province-year-gender level 
collapse work unemp inact active , by(provin year woman) 
	
	cap drop unemp_r /*proportion of active population that is unemployed */
	gen unemp_r=unemp/active
	
	keep provin year woman work inact unemp_r 
	label variable unemp_r "unemployment rate"
	label variable work "share of workers"
	
*Gaps between men and women
reshape wide work inact unemp_r, i(provin year) j(woman)

rename work0 work_men
rename work1 work_women
rename inact0 inact_men
rename inact1 inact_women
rename unemp_r0 unemp_men
rename unemp_r1 unemp_women

gen work_gap=work_men-work_women
gen inact_gap=inact_women-inact_men
gen unemp_gap=unemp_women-unemp_men

rename provin cprov


	label var work_men "Share of male workers"
	label var inact_men "Share of inactive men"
	label var unemp_men "Male unemployment rate"
	
	label var work_women "Share of female workers"
	label var inact_women "Share of inactive women"
	label var unemp_women "Female unemployment rate"

	label var work_gap "male workers-female workers"
	label var inact_gap "female inactive-male inactive"
	label var unemp_gap "female unemployment rate-male unemployment rate"
	
	label var cprov "Province"
	
save "Datasets/01_processeddata/epa_prov_year_gender.dta", replace

** Note: to produce Figure 4, we need to merge this dataset with our main dataset.
* This is done in "03_main_analysis_code.do"


